

Kreslení čar

© 1995-2001 Josef Pelikán
KSVI MFF UK Praha


e-mail: Josef.Pelikan@mff.cuni.cz

WWW: <http://cgg.ms.mff.cuni.cz/~pepca/>


Kreslení úseček



vektorové zařízení




rastrové zařízení



OK

DDA algoritmus



DDA algoritmus

```
procedure LineDDA ( x1, y1, x2, y2, color : integer );  
    { předpoklady: x1 < x2, |y2-y1| < |x2-x1| }  
var y, dy : real;  
begin  
    y := y1;  
    dy := (y2-y1) / (x2-x1);  
    PutPixel(x1,y1,color);  
    while x1 < x2 do  
        begin  
            x1 := x1 + 1;  
            y := y + dy;  
            PutPixel(x1,round(y),color);  
        end;  
end;
```

DDA algoritmus


+ **výhody:**

- snadná implementace (HW)

→ **nevýhody:**

- nutno počítat s velkou **přesností** (real, fixed point)
- jedno **dělení** a v cyklu **zaokrouhlování**

Bresenhamův algoritmus



$$dx = x_2 - x_1$$

$$dy = y_2 - y_1$$

$$E' = E + \frac{dy}{dx} \leq M = \frac{1}{2}$$

Celočíselné odvození

$$E' = E + \frac{dy}{dx} \leq \frac{1}{2} \quad / \cdot 2dx$$

$$2dx \cdot E' = 2dx \cdot E + 2dy \leq dx \quad / - dx$$

$$dx(2E' - 1) = dx(2E - 1) + 2dy \leq 0$$



$$D' = D + 2dy \leq 0$$

$$D_0 = 2dy - dx$$

$$D \leq 0 \Rightarrow D' = D + 2dy, \quad y' = y$$

$$D > 0 \Rightarrow D' = D + 2dy - 2dx, \quad y' = y + 1$$


Bresenhamův algoritmus

```
procedure LineBres ( x1, y1, x2, y2, color : integer );  
    { předpoklady: x1 < x2, |y2-y1| < |x2-x1| }  
var dx, dy, D, inc0, inc1 : integer;  
begin  
    dx := x2 - x1;  
    dy := y2 - y1;  
    D := 2*dy - dx;  
    inc0 := 2*dy;  
    inc1 := 2*(dy - dx);  
    PutPixel(x1,y1,color);  
    ...
```




Bresenhamův algoritmus

```
...  
while x1 < x2 do  
  begin  
    if D <= 0 then D := D + inc0  
    else  
      begin  
        D := D + inc1;  
        y1 := y1 + 1;  
      end;  
      x1 := x1 + 1;  
      PutPixel(x1,y1,color);  
    end;  
end;
```


Jednokrokový algoritmus




Víceřkové algoritmy



Kreslení kružnice




Kreslení kružnice



Kreslí se jen jedna osmina oblouku
- zbytek se přenesse pomocí symetrií

Bresenhamův algoritmus



$$F(M) = M_x^2 + M_y^2 - R^2$$

Inkrementální odvození

$$1) \quad F(M') = (x + 1)^2 + (y - \frac{1}{2})^2 - R^2 < 0$$

$$F(M'') = (x + 2)^2 + (y - \frac{1}{2})^2 - R^2 = F(M') + 2x + 3$$

$$2) \quad F(M') \geq 0$$

$$F(M'') = (x + 2)^2 + (y - \frac{3}{2})^2 - R^2 = F(M') + 2x - 2y + 5$$

$$D_0 = 1.25 - R \quad \{1 - R\}$$

$$D < 0 \Rightarrow D' = D + 2x + 3, \quad y' = y$$

$$D \geq 0 \Rightarrow D' = D + 2x - 2y + 5, \quad y' = y - 1$$

Kreslení kružnice

```
procedure CirclePoints ( x, y, color : integer );  
    { předpoklad: střed kružnice je v počátku }  
begin  
    PutPixel ( x, y, color );  
    PutPixel ( y, x, color );  
    PutPixel ( x, -y, color );  
    PutPixel ( y, -x, color );  
    PutPixel ( -x, y, color );  
    PutPixel ( -y, x, color );  
    PutPixel ( -x, -y, color );  
    PutPixel ( -y, -x, color );  
end;
```

...

Kreslení kružnice

```
procedure CircleBres ( R, color : integer );  
var x, y, D : integer;  
begin  
  x := 0; y := R; D := 1 - R;  
  CirclePoints(0,R,color);  
  while y > x do  
    begin  
      if D < 0 then D := D + 2*x + 3  
        else  
          begin  
            D := D + 2*(x - y) + 5;  
            y := y - 1;  
          end;  
          x := x + 1;  
          CirclePoints(x,y,color);  
        end;  
    end;  
end;
```

Konec

Další informace:

- **J. Foley, A. van Dam, S. Feiner, J. Hughes:** *Computer Graphics, Principles and Practice*, 72-87
- **Jiří Žára a kol.:** *Počítačová grafika*, principy a algoritmy, 91-100, 106-112
- ➔ **LAN na Malé Straně:**
 - **barbora\usr:\vyuka\pelikan\2**